Solution with Comments to Assignment 3: Integer Programming and Nonlinear Programming

There are five questions (15 total points) in this assignment. All relevant Excel files can be found on Sakai. Solve these questions in Excel and fill in the solution template provided below.

Question 1: Problem 43 on page 268 in the PMS 5th Ed textbook.

Step 1: Specify the Excel file Question1.xlsx. Make sure to record *all the necessary formulas*.

	A	B	C	D	E	F	G	H	I
1	**Traveling from New York to LA**								
2									
3	Labeling of nodes								
4	City	Index							
5	New York	1							
6	Cleveland	2							
7	St. Louis	3							
8	Nashville	4							
9	Phoenix	5							
10	Dallas	6							
11	Salt Lake City	7							
12	Los Angeles	8							
13									
14	Network formulation					Node balance constraints			
15	Origin	Destination	Gallons	Flow		Node	Net outflow (Outflow - Inflow)		Required net outflow
16	1	2	400	0		1	=SUMIF($A$16:$A$29,F16,$D$16:$D$29)-SUMIF($B$16:$B$29,F16,$D$16:$D$29)	=	1
17	1	3	950	1		2	=SUMIF($A$16:$A$29,F17,$D$16:$D$29)-SUMIF($B$16:$B$29,F17,$D$16:$D$29)	=	0
18	1	4	800	0		3	=SUMIF($A$16:$A$29,F18,$D$16:$D$29)-SUMIF($B$16:$B$29,F18,$D$16:$D$29)	=	0
19	2	5	1800	0		4	=SUMIF($A$16:$A$29,F19,$D$16:$D$29)-SUMIF($B$16:$B$29,F19,$D$16:$D$29)	=	0
20	2	6	900	0		5	=SUMIF($A$16:$A$29,F20,$D$16:$D$29)-SUMIF($B$16:$B$29,F20,$D$16:$D$29)	=	0
21	3	5	1100	1		6	=SUMIF($A$16:$A$29,F21,$D$16:$D$29)-SUMIF($B$16:$B$29,F21,$D$16:$D$29)	=	0
22	3	6	600	0		7	=SUMIF($A$16:$A$29,F22,$D$16:$D$29)-SUMIF($B$16:$B$29,F22,$D$16:$D$29)	=	0
23	4	6	600	0		8	=SUMIF($A$16:$A$29,F23,$D$16:$D$29)-SUMIF($B$16:$B$29,F23,$D$16:$D$29)	=	-1
24	4	7	1200	0					
25	5	8	400	1					
26	6	5	900	0					
27	6	7	1000	0					
28	6	8	1300	0					
29	7	8	600	0					
30									
31	Gallons used	=SUMPRODUCT(C16:C29,D16:D29)							

Step 2: Specify Solver

Set Objective: B31

To: ○ Max X Min ○ Value of: _____

By Changing Variable Cells: D16:D29

Subject to the Constraints:

G16:G23=I16:I23
D16:D29=binary

□ Make Unconstrained Variables Non-Negative

Select a Solving Method: Simplex LP

Step 3: Report your results below.

The best route is New York (1) → St. Louis (3) → Phoenix (5) → Los Angeles (8).

Question 2: Problem 49 on page 269 in the PMS 5th Ed textbook.

Step 1: Specify the Excel file Question2.xlsx. Make sure to record *all the necessary formulas*.

Hint: Hours Spent cannot go beyond Hours Required.

	A	B	C	D	E	F	G	H	I	J
1	**Touche Young auditors**									
2										
3	Amount billed per hour									
4		Project								
5	Auditor	1	2	3	4	5	6			
6	1	160	130	130	190	160	150			
7	2	170	140	170	160	160	180			
8	3	130	170	160	170	160	160			
9	4	180	190	130	190	170	190			
10	5	130	140	170	130	130	170			
11	6	140	160	170	150	150	170			
12	7	150	180	140	130	140	140			
13	8	150	170	190	160	120	140			
14										
15	Hours spent									
16		Project								
17	Auditor	1	2	3	4	5	6	Auditor hours		Available
18	1	0	0	0	160	0	0	=SUM(B18:G18)	<=	160
19	2	60	0	0	0	0	100	=SUM(B19:G19)	<=	160
20	3	0	0	0	10	150	0	=SUM(B20:G20)	<=	160
21	4	120	40	0	0	0	0	=SUM(B21:G21)	<=	160
22	5	0	0	40	0	0	90	=SUM(B22:G22)	<=	160
23	6	0	0	0	0	0	0	=SUM(B23:G23)	<=	160
24	7	0	160	0	0	0	0	=SUM(B24:G24)	<=	160
25	8	0	0	160	0	0	0	=SUM(B25:G25)	<=	160
26	Hours spent	=SUM(B18:B25)	=SUM(C18:C25)	=SUM(D18:D25)	=SUM(E18:E25)	=SUM(F18:F25)	=SUM(G18:G25)			
27		<=	<=	<=	<=	<=	<=			
28	Hours required	180	200	200	170	150	190			
29										
30	Total billings	=SUMPRODUCT(B6:G13,B18:G25)								

Step 2: Specify Solver
Set Objective: B30
To: X Max ○ Min ○ Value of: _____
By Changing Variable Cells: B18:G25
Subject to the Constraints:

B26:G26<=B28:G28 ("=" would work too.)
H18:H25<=J18:J25

X Make Unconstrained Variables Non-Negative
Select a Solving Method: Simplex LP

Comments:

- Many students set B26:G26>=B28:G28. → -0.5 "As stated in the hint, hours spent cannot go beyond hours required. Otherwise all auditors will work 160 hours."

Step 3: Report your results below.
The maximal total billings = 194,800.

Question 3: Problem 43 on page 328 in the PMS 5th Ed textbook.

Step 1: Specify the Excel file Question3.xlsx. Make sure to record all the necessary formulas.

	A	B	C	D	E	F	G	H	I	J
1	**Basketball lineup**									
2										
3	Data on players									
4		Player 1	Player 2	Player 3	Player 4	Player 5	Player 6	Player 7		
5	Guard	1	0	1	0	1	0	1		
6	Forward	0	0	1	1	1	1	1		
7	Center	0	1	0	1	0	1	0		
8	Ball-handling	3	2	2	1	1	3	3		
9	Shooting	3	1	3	3	3	1	2		
10	Rebounding	1	3	2	3	1	2	2		
11	Defense	3	2	2	1	2	3	1		
12										
13		Player 1	Player 2	Player 3	Player 4	Player 5	Player 6	Player 7	Total	
14	Player plays	1	1	1	0	1	0	1	=SUM(B14:H14)	=
15										
16	Constraints on positions									
17		Playing		Required						
18	Guard	=SUMPRODUCT(B5:H5,$B$14:$H$14)	>=	4						
19	Forward	=SUMPRODUCT(B6:H6,$B$14:$H$14)	>=	2						
20	Center	=SUMPRODUCT(B7:H7,$B$14:$H$14)	>=	1						
21										
22	Skill constraint (interpreted as requiring average of all three, ball-handling, shooting, and rebounding, to be at least 2)									
23		Average		Required						
24	Ball-handling	=SUMPRODUCT(B8:H8,$B$14:$H$14)/$K$14	>=	1.8						
25	Shooting	=SUMPRODUCT(B9:H9,$B$14:$H$14)/$K$14	>=	1.8						
26	Rebounding	=SUMPRODUCT(B10:H10,$B$14:$H$14)/$K$14	>=	1.8						
27										
28	At least one of player 2 or 3 must start									
29		Sum of 2,3		Required						
30		=SUM(C14:D14)	>=	1						
31										
32	Total defense	=SUMPRODUCT(B11:H11,B14:H14)								

Comments:

- For the average of ball-handling, shooting, and rebounding, "The Sumproduct function gives you the total, which needs to be divided by 5."

Step 2: Specify Solver

Set Objective: B32

To: X Max ○ Min ○ Value of: _____

By Changing Variable Cells: B14:H14

Subject to the Constraints:

B14:H14=binary
B18:B20>=D18:D20
B24:B26>=D24:D26
B30>=D30
I14=K14

X Make Unconstrained Variables Non-Negative

Select a Solving Method: Simplex LP

Comments:

- Many students used GRG Nonlinear. → "In our class, we only covered Simplex LP and GRG Nonlinear methods. If your model doesn't contain any nonlinear function, then we prefer to use the Simplex LP method."

Step 3: Report your results below.

The optimal total defensive ability of the starting team = 10.

Question 4: Problem 70 on page 331 in the PMS 5th Ed textbook.

Notes: Suppose the capacity of each shift is 10,000. Then the effective capacity of each shift is either 10,000 if the shift is run, or 0 if the shift is not run.

Part a.

Step 1: Use B13:E13,B15:E15 as decision variables. Specify the Excel file Question4.xlsx. Make sure to record all the necessary formulas.

	A	B	C	D	E
1	**Production scheduling with shifts**				
2					
3		Day	Night		
4	Setup cost	8000	4500		
5					
6	Unit holding cost	1			
7					
8	Demands (in thousands)				
9		Day 1	Night 1	Day 2	Night 2
10	Demand	2000	3000	2000	3000
11					
12		Day 1	Night 1	Day 2	Night 2
13	Run shift	1	1	0	1
14					
15	Quantity produced	2000	5000	0	3000
16		<=	<=	<=	<=
17	Effective capacity	=B13*10000	=C13*10000	=D13*10000	=E13*10000
18					
19	Ending inventory	=B15-B10	=B19+C15-C10	=C19+D15-D10	=D19+E15-E10
20		>=	>=	>=	>=
21	Meet demand on time	0	0	0	0
22					
23	Summary of costs:				
24	Total setup cost	=SUMPRODUCT(B4:C4,B13:C13)+SUMPRODUCT(B4:C4,D13:E13)			
25	Total holding cost	=B6*SUM(B19:E19)			
26	Total cost	=SUM(B24:B25)			

Step 2: Specify Solver

Set Objective: B26

To: ○ Max X Min ○ Value of: _____

By Changing Variable Cells: B13:E13,B15:E15

Subject to the Constraints:

B13:E13=binary
B15:E15<=B17:E17
B19:E19>=B21:E21

X Make Unconstrained Variables Non-Negative

Select a Solving Method: Simplex LP

Step 3: Report your results below.

Optimal production schedule:

	Day 1	Night 1	Day 2	Night 2
Run shift	1	1	0	1
Quantity produced	2,000	5,000	0	3,000

The optimal total cost = 19,000.

Part b. Report your updated results below.

Optimal production schedule:

	Day 1	Night 1	Day 2	Night 2
Run shift	1	0	1	0
Quantity produced	5,000	0	5,000	0

The optimal total cost = 8,000.

Is it reasonable that the decrease in setup costs has actually raised the average inventory level? Explain.
Yes, it is reasonable. The day shift setup decrease is so large that it is now best to produce only during day shifts, even though this entails more inventory carried from day to night.

Question 5: Problem 62 on page 403 in the PMS 5th Ed textbook.

Step 1: Specify the Excel file Question5.xlsx. Make sure to record all the necessary formulas.

	A	B	C
1	**Locating a fire station**		
2			
3	Locations of towns		
4		x	y
5	Town 1	10	20
6	Town 2	60	20
7	Town 3	40	30
8	Town 4	80	60
9			
10	Numbers of fires per year		
11	Town 1	20	
12	Town 2	30	
13	Town 3	40	
14	Town 4	25	
15			
16		x	y
17	Location of fire station	40.0000000058144	29.9999420786202
18			
19	"City-block" distances from fire station to towns		
20	Town 1	=ABS(B5-$B$17)+ABS(C5-$C$17)	
21	Town 2	=ABS(B6-$B$17)+ABS(C6-$C$17)	
22	Town 3	=ABS(B7-$B$17)+ABS(C7-$C$17)	
23	Town 4	=ABS(B8-$B$17)+ABS(C8-$C$17)	
24			
25	Average distance to a fire	=SUMPRODUCT(B11:B14,B20:B23)/SUM(B11:B14)	

Step 2: Specify Solver
Set Objective: B25

To: ○ Max X Min ○ Value of: _____
By Changing Variable Cells: B17:C17
Subject to the Constraints:

(No constraints)

☐ Make Unconstrained Variables Non-Negative
Select a Solving Method: GRG Nonlinear

Comments:

- Some students specified the integer or nonnegative constraints. But they are not necessary.
- Some students calculated the average distance to a fire wrong. Some of them directly used average.
- This is an unconstrained problem.

Step 3: Report your results below.

	x	y
Location of fire station	40	30

The optimal average distance to a fire = 30.